

Feedline Data Calculations for the VZ200/300

Rick Buhre VK4AI
41 Mogford Street, Mackay, Qld. 472

This program came about when the price of the VZ200 dropped dramatically

The story of how this program came about is simple, but I believe it could be of interest. It all began when the price dropped on the VZ200 and Wal VK4AIV, bought one.

After learning the basics of its operation, he began to search for useful programs involving amateur radio, finding them few and far between.

Much later, I purchased a VZ300 at the same price as Wal's VZ200 and naturally asked Wal what programs he had.

Upon discovering the scarcity, I sat down and wrote a series of short programs to ease the problems of endless work with calculator,

pen and paper, for amateur radio work.

Copies of these programs were given to Wal, who tidied them up and tied them together. This listing is part of the result.

The program is to enable those interested to quickly calculate parameters for the construction of coaxial cable or open wire feeder sections for matching antennas to feedlines.

The calculations are derived from standard amateur radio books and simply are converted into Basic statements.

They are as follows:

COAXIAL CABLE DATA

- 1 Impedance of a cable of a given size.

- 2 Inside diameter of outer conductor for given impedance and inner conductor size
- 3 Outside diameter of inner conductor for given impedance and outer conductor size
- 4 Cut off frequency for a cable of given size and impedance.

OPEN WIRE FEEDER DATA

- 1 Impedance of feeders of known wire size and spacing.
- 2 Spacing required for a given wire size and impedance.

There is space in the program for future additions to be inserted. I hope many amateurs will find it of use.

```

10 CLS:GOSUB3000
20 PRINT@99,"1- COAXIAL CABLE DATA"
30 PRINT@195,"2- OPEN WIRE FEEDER DATA "
40 PRINT@291,"3- "
50 PRINT@387,"4- "
60 PRINT@448,"CHOOSE OPTION":INPUTN
70 IFN=1THEN100
80 IFN=2THEN2000
85 REM*****
90 REM*****
100 GOSUB3000
110 PRINT@99,"1-IMPEDANCE OF COAXIAL."
120 PRINT@131,"      CABLE"
130 PRINT@195,"2-INSIDE DIA.OF OUTER"
140 PRINT@227,"      CONDUCTOR"
150 PRINT@291,"3-OUTSIDE DIA.OF INNER"
160 PRINT@387,"4-CUT OFF FREQUENCY"
170 PRINT@448,"CHOOSE OPTION":INPUTN
180 IFN=1THEN500
190 IFN=2THEN1000
200 IFN=3THEN1200
210 IFN=4THEN1400
220 IFN<1THEN1010
230 IFN>4THEN1010
235 REM*****
240 REM*****
500 GOSUB2500
510 INPUT"ENTER INSIDE DIAMETER OF OUTER CONDUCTOR"/D1
520 INPUT"ENTER OUTSIDE DIAMETER OF INNER CONDUCTOR"/D0
530 X=SQR(K)
540 Y=D1/D0
550 Z=LOG(Y)/2.30259
560 W=138*X/Z
570 PRINTW;"OHMS IMPEDENCE"

```



```

580 PRINT"ANOTHER TRY?Y,N"
590 INPUTA$
600 IFA$=CHR$(89)THEN500
610 IFA$=CHR$(78)THEN10
620 REM*****
630 REM*****
1000 GOSUB2500
1010 INPUT"ENTER IMPEDANCE";Z
1020 INPUT"ENTER OUTSIDE DIAMETER OF INNER CONDUCTOR";D
1030 X=SQR(K):Y=Z*X/138
1040 W=(10^Y)*D
1050 PRINT"INSIDE DIAMETER OF OUTER CONDUCTOR=";W
1060 PRINT"ANOTHER TRY?Y,N"
1070 INPUTA$
1080 IFA$=CHR$(89)THEN1000
1090 IFA$=CHR$(78)THEN10
1091 REM*****
1092 REM*****
1200 GOSUB2500
1210 INPUT"ENTER IMPEDANCE";Z
1220 INPUT"ENTER INSIDE DIAMETER OF OUTER CONDUCTOR";D
1230 T=SQR(K)
1240 U=Z*T/138
1250 V=10^U
1260 W=1/V
1280 X=W*D
1290 PRINT"OUTSIDE DIAMETER OF INNER CONDUCTOR=";X
1300 PRINT"ANOTHER TRY?Y,N"
1310 INPUTA$
1320 IFA$=CHR$(89)THEN1200
1330 IFA$=CHR$(78)THEN10
1390 REM*****
1391 REM*****
1400 GOSUB2500
1410 INPUT"ENTER INSIDE DIA. OUTER CONDUCTOR";D1
1420 INPUT"ENTER OUTSIDE DIA. INNER CONDUCTOR";D0
1430 Z=SQR(K)
1440 X=7520/(D1+D0)*Z
1450 PRINT"CUT OFF FREQUENCY=";X;"MHZ"
1460 PRINT"ANOTHER TRY?Y,N"
1470 INPUTA$
1480 IFA$=CHR$(89)THEN1400
1490 IFA$=CHR$(78)THEN10
1491 REM*****
1492 REM*****
1516 REM*****
2000 GOSUB3000
2010 PRINT@99,"1-IMPEDANCE OF OPEN"
2020 PRINT@131,"      WIRE FEEDER"
2030 PRINT@195,"2-SPACING OF OPEN"
2040 PRINT@227,"      WIRE FEEDER"
2050 PRINT@291,"3- "
2060 PRINT@387,"4-"
2070 PRINT@449,"CHOOSE OPTION":INPUTH
2080 IFH=2THEN2400
2100 IFH=1THEN2200
2110 A$=INKEY$:IFA$<>CHR$(45)THEN2110
2120 IFA$=CHR$(45)THEN10

```

```

2191 REM*****
2200 CLS:PRINT"OPEN WIRE IMPEDANCE"
2210 INPUT"SPACING";D1
2220 INPUT"DIA OF WIRE";D2
2230 X=D1/D2
2240 W=X+SQR((X*X)-1)
2250 Y=LOG(W)/2.30259
2260 Z=Y*276
2270 PRINTZ;"OHMS IMPEDANCE"
2280 PRINT"ANOTHER TRY?Y,N"
2290 INPUTA$
2300 IFA$=CHR$(89)THEN2200
2310 IFA$=CHR$(78)THEN10
2400 CLS:PRINT"TO FIND SPACING OPEN WIRE"
2410 INPUT"ENTER IMP";Z0
2420 INPUT"WIRE DIA";D
2430 X=Z0/276:Y=10^X:A=D*(Y*Y-1):S=A/(2*Y):PRINT"SPACING=";S
2440 PRINT"ANOTHER TRY?Y,N"
2450 INPUTA$
2460 IFA$=CHR$(89)THEN2400
2470 IFA$=CHR$(78)THEN10
2500 CLS:PRINT:PRINT"DIELECTRIC CONSTANTS:"PRINT"AIR=1"
2510 PRINT"POLYTHENE=2.26":PRINT"FOAM POLYTHENE=1.2"
2520 PRINT"TEFLON=2.1"
2560 INPUT"ENTER DIELECTRIC CONSTANT";K
2570 RETURN
2571 REM*****
3000 CLS:PRINT00," *****"
3010 PRINT032," *          +++++MENU+++++          *"
3020 PRINT064," *                                     *"
3030 PRINT096," *                                     *"
3040 PRINT128," *                                     *"
3050 PRINT160," *                                     *"
3060 PRINT192," *                                     *"
3070 PRINT224," *                                     *"
3080 PRINT256," *                                     *"
3090 PRINT288," *                                     *"
3100 PRINT320," *                                     *"
3110 PRINT352," *                                     *"
3120 PRINT384," *                                     *"
3130 PRINT416," *****"
3150 RETURN

```